

Supplement Document

OpenBUGS data analysis code for Model 2 (Proposed model)

```
Model
{
  for( i in 1 : N ) {
    for( j in 1 : T ) {
      Y[i , j] ~ dnorm(theta[i ,j],tauc[i])
      mu[i , j] <- alpha[i] + beta[i]*step(x[j]-CP[i])+ ca[i] * (x[j]) +
      da[i]*(x[j] - CP[i])*step(x[j]-CP[i])
    }
    theta [i,1]<- mu [i,1]
    for ( j in 2 : T ) {
      theta[i ,j]<-mu[i ,j]+tgamma[i]*(Y[i ,j-1]-mu[i ,j-1])
    }
    alpha[i] ~ dnorm(alphac,alphatau)
    beta[i] ~ dnorm(betac,betatau)
    ca[i] ~ dnorm(cac,catau)
    da[i] ~ dnorm(dac,datau)
    tgamma[i]~dnorm(simge,gr)I(-0.99999,0.99999)
    tsigma[i] ~ dunif(sa,sb)
    tauc[i] <- 1 / (tsigma[i]*tsigma[i])
  }
  alphac ~ dnorm(0.0,1.0E-6)
  betac ~ dnorm(0.0,1.0E-6)
  cac ~ dnorm(0.0,1.0E-6)
  dac ~ dnorm(0.0,1.0E-6)
  sigmaalpha~ dunif(0,100)
  sigmabeta~ dunif(0,100)
  sigmaca~ dunif(0,100)
  sigmada~ dunif(0,100)
  alphatau<-pow(sigmaalpha, -2)
  betatau<-pow(sigmabeta, -2)
  catau<-pow(sigmaca, -2)
  datau<-pow(sigmada, -2)
  simge~dnorm(0.0,1.0E-6)
  simgr~ dunif(0,100)
  gr <- pow(simgr, -2)
  sa~ dunif(0,100)
  sb~ dunif(sa,100)
  tmsig<-mean(tsigma[])
  tmgamma<-mean(tgamma[])
  smsig<- (sa+sb)/2
  svsig<- sqrt((pow((sb-sa), 2))/12)
}
```